
Mithilfe des Worksheet-Objekts, das eine Tabelle reprasentiert, konnen Sie auf eine bzw.
auch auf alle Tabellen einer Arbeitsmappe tiber die Worksheets-Auflistung zugreifen. In der
Auflistung Worksheets sind standardmifSig alle Tabellen einer Arbeitsmappe verzeichnet.

Sehen Sie sich einmal Abbildung 4.1 an, um einen ersten Eindruck von der Befehlsvielfalt
des Objekts Worksheet zu bekommen.

TABELLE

Eine Tabelle

Tabellel
Worksheets("Tabellel")
Sheets("Diagramm")
Worksheets(1)

Mehrere Tabellen
Sheets(Array("Tabellel", "Tabelle2"))

Methoden/ Eigenschaften
Add (M)

AutoFilter (E)
AutoFilterMode (E)
Calculate (M)

CodeName (E)

Copy (M)

Count (E)

Delete (M)
ExportAsFixedFormat (M)
Index (E)

Move (M)

Name (E)

PageSetup (E)
PivotTables (M)
PrintOut (M)

Protect/ Unprotect (M)
ResetAl1PageBreaks (M)
ScrollArea (E)
ShowAl1Data (M)

Tab (E)

UsedRange (E)

Visible (E)

Abbildung 4.1: Die wichtigsten Methoden und Eigenschaften des Objekts Worksheet.

Im oberen Bereich von Abbildung 4.1 sehen Sie verschiedene Moglichkeiten, eine Tabelle
anzusprechen. Wir haben dies bereits im ersten Kapitel dieses Buchs besprochen. Im unte-
ren Bereich der Abbildung sehen die wichtigsten Eigenschaften (E) und Methoden (M) fur
dieses Objekt.

Die folgenden Beispiele sind in der Arbeitsmappe Start.xlsm im Modul md|_Tabellen
§—> zu finden.

Lernen Sie auf den nichsten Seiten, wie Sie das Objekt Worksheet in der Praxis einsetzen
kénnen.

Tabellen dokumentieren, filtern und durchsuchen

Auf den nichsten Seiten erfahren Sie, wie Sie unter anderem ein Inhaltsverzeichnis aller
Tabellen einer Arbeitsmappe erstellen und verlinken kénnen. Des Weiteren werden Sie
alle Tabellen einer Arbeitsmappe nach Formeln und Verkniipfungen durchsuchen und die
Fundstellen dokumentieren.

Tabellen dokumentieren, filtern und durchsuchen

Tabelleninhaltsverzeichnis erstellen und verlinken

Das erste Makro in diesem Kapitel ist schon der Knaller. Es erstellt in der Tabelle tbl_Ta-
bellen ein Inhaltsverzeichnis aller Tabellen der Mappe und hinterlegt sogar fiir jeden Ein-
trag einen Hyperlink.

Sub TabellenListenUndInhaltsverzeichnisErstellen()
Dim Blatt As Worksheet
Dim Zeile As Long

With tbl Tabellen
.UsedRange.Clear
Zeile =1

For Each Blatt In Worksheets

.Range("A" & Zeile).Value = Blatt.Name
.Hyperlinks.Add Anchor:=.Range("A" & Zeile), _
Address:="", _
SubAddress:="'" & Blatt.Name & "'!Al"
Zeile = Zeile + 1

Next Blatt
End With

End Sub

Listing 4.1: Ein Inhaltsverzeichnis aller Tabellen in Sekundenschnelle erstellt.

Zu Beginn des Makros aus Listing 4.1 deklarieren wir eine Variable vom Typ Worksheet mit
dem Namen Blatt. Damit gewinnen wir automatisch Zugriff auf alle Methoden und
Eigenschaften dieses Objekts. Die Variable zeile vom Typ Long dient spiter zur Verwaltung
der Zeilen in der Tabelle tbl_Tabellen. Nachdem wir mit der Anweisung With erklirt
haben, in welcher Tabelle wir das Inhaltsverzeichnis anlegen mochten, wenden wir die
Methode Clear auf den benutzten Bereich dieser Tabelle an, um den Bereich sicherheitshal-
ber vorher zu 16schen. Danach setzen wir den Inhalt der Variablen zeile auf den Wert 1.
Damit legen wir fest, dass der erste Eintrag ab der ersten Zeile dieser Tabelle geschrieben
werden soll.

Mit einer For Each...Next-Schleife, die Sie im zweiten Kapitel, »Die wichtigsten Sprachele-
mente von Excel-VBA«, kennengelernt haben, arbeiten wir uns Tabelle fiir Tabelle in der
aktuellen Arbeitsmappe vor und schreiben die Namen der Tabelle mithilfe der Eigen-
schaft Name hiibsch untereinander in die Tabelle tbl_Tabellen.

Danach wenden wir die Methode Add an, um in der jeweiligen Zelle einen Hyperlink ein-
zufiigen. Im Parameter Anchor geben Sie den sogenannten Anker (die Zelle) des Hyperlinks
an. Der Parameter Address bleibt fiir dieses Beispiel leer. Im letzten Parameter SubAddress
geben Sie das Sprungziel des Hyperlinks an. Dieses Sprungziel besteht aus dem eigentli-
chen Namen der Tabelle sowie der gewiinschten Zellenkoordinate.

Wir diirfen nicht vergessen, die Zihlvariable Zeile nach jedem Schreibvorgang um den
Wert 1 zu erhohen, damit der nichste Tabellenname jeweils in die nidchste Zeile geschrie-
ben werden kann.

A7 M f || tbl Export v

Tabellel
Tabelle2

tbl Zellen

tbl Prifen

tbl Schleife

tbl Datumsreihe
thl Export |
tbl Farben

tbl Konvertieren
tbl Suchen

tbl Direktsuche
tbl Wverweis
tbl Extakt

tbl Haupt
15 |tbl Neben

16 .

« ... | tblBereiche | tblBereiche2 | tbl Tabellen @ [« |

Abbildung 4.2: Alle Tabellen der Mappe werden untereinander aufgelistet.

Kennen Sie den Unterschied zwischen den Befehlen Worksheets und Sheets? Uber
— den Befehl Worksheets(1) kdnnen Sie das erste Tabellenblatt einer Arbeitsmappe
ansprechen, mit Sheets(1) sprechen Sie ebenfalls das erste Blatt in der Arbeits-
mappe an.
Aber: Sheets(1) kann genauso gut auch ein Diagrammblatt sein, Worksheets (1) ist
jedoch immer nur eine Excel-Tabelle. Da es fir Diagramme andere Befehle als fiir
normale Tabellen gibt, ist die Verwechslung beider Begriffe etwas gefahrlich. Ver-
wenden Sie also Worksheets, wenn Sie Tabellen ansprechen méchten!

Tabellen durchsuchen und dokumentieren

In der folgenden Aufgabe wird die komplette Arbeitsmappe nach Formeln und Verkniip-
fungen durchsucht. Die dabei gefundenen Zellen sollen im Direktfenster der Entwick-
lungsumgebung dokumentiert werden. Starten Sie zu diesem Zweck das Makro aus Lis-
ting 4.2.

Sub FormelKundschafter()
Dim Blatt As Worksheet
Dim zelle As Range

'AuBere Schleife.
For Each Blatt In Worksheets

For Each zelle In Blatt.UsedRange

If zelle.HasFormula Then
Debug.Print Blatt.Name & " /" & zelle.Address & " / " & zelle.Formulalocal
End If

Next zelle
Next Blatt

End Sub

Listing 4.2: Alle Formeln werden inklusive ihrer Adresse im Direktfenster ausgegeben.

Zu Beginn des Makros deklarieren wir zwei Objektvariablen. Mithilfe der Variablen Blatt
vom Typ Worksheet basteln wir die dufSere For Each...Next-Schleife, die Tabelle fir Tabelle in
der Arbeitsmappe verarbeitet. Uber den Einsatz der Variablen zelle vom Typ Range verar-
beiten wir alle Zellen des benutzten Bereichs und setzen dazu eine weitere For Each...Next-
Schleife ein.

Uber die Eigenschaft HasFormula kénnen wir priifen, ob die jeweilige Zelle eine Formel ent-
hilt. In diesem Fall geben wir den Namen der Tabelle iiber die Eigenschaft Name, die Koor-
dinate tiber die Eigenschaft Address sowie den Formeltext tiber die Eigenschaft Formulalocal
im Direktfenster der Entwicklungsumgebung aus.

tbl Duplikate /$B$13 / —ZAHLENWENN (A:R;A13) -
tbl Duplikate /$B$14 / —ZAHLENWENN (A:R;A14)
tbl Duplikate /$B$15 / =ZAHLENWENN (A:A;R15)
FormelnUndWerte /$B$5 / =B4-B3
FormelnUndWerte /$C$5 / =C4-C3
FormelnUndWerte /$D$5 / =D4-D3
FormelnUndWerte /$E$5 / =E4-E3
FormelnUndWerte /$F$5 / =F4-F3
FormelnUndWerte /$G$5 / =G4-G3 —

| =
< | »

Abbildung 4.3: Namen der Tabelle, Koordinate und Formeltext wurden ausgelesen.

Tabellen filtern mit einem Kriterium

Sicher kennen Sie den Datenfilter in Excel, iiber den Sie bestimmte Zeilen einer Tabelle fil-
tern konnen. Im folgenden Beispiel sollen in der Tabelle tbl_Filter alle Zeilen, die in Spalte
A die Region Siid haben, gefiltert werden. Sehen Sie sich zunichst die Ausgangssituation
in Abbildung 44 an.

E21 ~ Jx @

A A | B | C | D | E | F | ]

1 [Region Monat Umsatz

2 |Nord 1 49.043

3 |sud 1 22.996

4 |West 1 42,317

5 |Ost 1 35.480

6 |Nord 2 38.956

7 |sud 2 37.750

& |West 2 31.488

9 |Ost 2 46.448

10 |Nord 3 24.723

11 |sud 3 35.693

12 |West 3 23.406

13 |Ost 3 46.276

14 |Nord 4 20.384

15 |sud 4 17.860

16 |West 4 23.414 |

17 |Ost 4 19.718

18] -
« ... | tbl_Filter ® [ ] v

Bereit 3 E O -——&F——+ 100%

Abbildung 44: Eine Umsatzliste nach Region und Monat.
Setzen Sie den Datenfilter wie in Listing 4.3 gezeigt ein.
Sub AutofilterRegionFiltern()
With tbl _Filter
If Not .AutoFilterMode = True Then
.Range("A1") .AutoFilter
End If
.Range("A1") .AutoFilter Field:=1, Criterial:="=Sid"

End With

End Sub

Listing 4.3: Alle Zeilen mit der Region Siid werden gefiltert und angezeigt.

Mithilfe der Eigenschaft AutoFilterMode konnen Sie iiberpriifen, ob Filterpfeile in der
Tabelle iiberhaupt angezeigt werden. Falls nicht, wenden Sie die Methode AutoFilter an,
um die Filterpfeile einzublenden. Stellen Sie danach den gewiinschten Filter ein und tiber-
geben Sie im Parameter Field die Spaltennummer und im Parameter Criteriall das ge-

wiinschte Auswahlkriterium.

c20 -~ fe

4 A | s [N o | E | F |
1 |Region @Monat |Z|Umsalz |Z|

3 |sud 1 22,996

7 |sud 2 37.750

11 [sud 3 35.693

15 |sid a 17.860

18|

19

—
... | thlFilter ® : [ | [

Ed 0 -—F——+ 100%

s

Bereit

Abbildung 4.5: Alle Datenstze mit der Region Siid werden angezeigt.

Tabellen filtern mit mehreren Kriterien

Bei der nichsten Aufgabenstellung wird ein Filter angewendet, der zwei Kriterien erfiillen
muss. So sollen alle Zeilen gefiltert werden, die im Wertebereich zwischen 20000 und

30000 liegen. Dazu erfassen Sie das Makro aus Listing 44.

Sub AutoFilterMehrereBedingungen ()

With tbl Filter
If Not .AutoFilterMode = True Then
.Range("A1").AutoFiTlter
End If

.ShowAl1Data
.Range("A1") .AutoFilter Field:=3, Criterial:=">=20000", _
Operator:=x1And, Criteria2:="<=30000"

.Range("A1").Sort Keyl:=.Range("C1"), Orderl:=x1Ascending, Header:=x1Yes

End With

End Sub
Listing 44: Alle Umsdtze im Bereich von 20000 bis 30000 regionunabhdngig filtern und sortieren.

Nach erfolgter Priffung des Filterstatus mithilfe der Eigenschaft AutoFilterMode wird die
Methode ShowAl1Data eingesetzt, um gegebenenfalls alle Zeilen in der Tabelle anzuzeigen.
AnschlieRend wenden wir die Methode AutoFilter an und tibergeben dieser zwei Kriterien
iiber die Parameter Criterial und Criteria2. Da ein Von-bis-Wertebereich abgefragt wird,
verwenden wir als Operator das Schliisselwort x1And. Direkt im Anschluss sortieren wir
iiber die Methode Sort den Bereich auf Basis der Spalte C absteigend.

Seit Excel-Version 2007 gibt es zwar eine neue Sort-Methode, da diese neuere
= Methode aber leider den Nachteil hat, dass sie in friheren Versionen nicht einsetz-

bar ist und uns fir die normale Sortierung keinerlei Vorteile bringt, habe ich mich
entschieden, eine lauffahige Methode zu beschreiben, die wirklich in allen Excel-

Versionen funktioniert.

B19 e ¥
4 < | D | E | F
1 _[Region E Umsatz
3 |Nord 4 20.384
10 |sod 1 22.996
12 | west 3 23.406
14 | west 4 23.414
16 |Nord 3 24.723
18 |
v 2 -
“ ... | thl Filter ® [« ] [v]
Bereit H O -——F——+ 100%

Abbildung 4.6: Filtern und anschliefSend sortieren in einem Arbeitsgang.

Tabellen einrichten und schiitzen

Auf den nichsten Seiten geht es rund um das Thema Schutz von Excel-Tabellen und um
die Vorbereitung darauf. Sie erfahren, wie Sie Bildlaufbereiche einrichten, Tabellenberei-
che sperren und Tabellen schiitzen.

Bildlaufbereiche fiir Tabellen festlegen

Im ersten Kapitel des Buchs sind wir bereits mit der Eigenschaft Scro11Area in Berithrung
gekommen. Dabei haben Sie erfahren, dass Sie den Bildlaufbereich einer Tabelle, also den
Bereich einer Tabelle, in dem sich der Anwender authalten darf, zwar iiber das Eigenschatf-
tenfenster setzen, aber leider eben nicht dauerhaft einstellen kénnen.

Gliicklicherweise konnen Sie diese Einstellung tiber das Ereignis Workbook Open auch dau-
erhaft vornehmen. Fithren Sie zur Festlegung des Codes, der bei Eintreten dieses Ereignis-
ses ausgefiihrt werden soll, im Projekt-Explorer einen Doppelklick auf die Rubrik Diese-
Arbeitsmappe aus und wihlen Sie im linken Drop-down-Listenfeld tiber dem Codefenster
das Objekt Workbook aus. Erginzen Sie danach den Ereigniscode wie in Listing 4.5 gezeigt.

Private Sub Workbook Open()

'ScrollArea fiir einen bestimmten Bereich einer Tabelle einstellen

tb1 ErsteMakros.ScrollArea = "Al:A10"
End Sub
Listing 4.5: Das Ereignis Open wird automatisch beim Offnen der Mappe ausgefiihrt.

Geben Sie der Eigenschaft ScrollArea bekannt, wo der Aufenthaltsbereich in der Tabelle
eingestellt werden soll.

Erfahren Sie mehr Uber den Einsatz von Ereignissen in Kapitel 7, »Die Ereignispro-
v grammierung in Excel.

Bereiche in Tabellen sperren

Eine manuell eher etwas aufwendigere Aufgabe ist das Sperren von einzelnen Zellen, die
Formeln enthalten. Viele Kalkulationen gehen schief, wenn versehentlich Formeln tiber-
schrieben oder geloscht werden. Daher konnen Sie ganz gezielt bestimmte Zellen fiir eine
Eingabe sperren. Dieser Sperrmechanismus ist jedoch standardméfRig nur in Verbindung
mit dem Tabellenschutz moglich. Zur Vorbereitung auf den eigentlichen Schutz der
Tabelle bestimmen Sie in der Tabelle tbl_FormelnUndWerte die Zellen, die Sie als nicht
tberschreibbar kennzeichnen mochten.

D9 g Je v
4 A | B | ¢ | E | F | 6 | H
1]
2 | Januar Februar Mérz April Mai Juni
3 |Kosten 14.707 10.575 11.348 11.711 13.895 13.761
4 |Ertrige 10.155 11.509 13.566 12.972 18.751 21.918
5 |Ergebnis -4.552 934 2.218 1.261 4.856 8.157
6 |
7
8 -
==
« v .| tblFormelnUndWerte | .. @ : [{] I
Bereit 25 H O -——F——+ 100%

Abbildung 4.7: Eine Monatsdarstellung von Kosten und Ertrdgen.

In Abbildung 4.7 sehen Sie einige Zellen, die mit dem Schriftschnitt Fett formatiert sind.
Genau diese Zellen sollen zukiinftig geschiitzt werden. Um sie vor einer versehentlichen
Loschung zu schiitzen, miissen Sie den Zellen das Merkmal Gesperrt tiber das Dialogfeld
Zellen formatieren/Schutz zuweisen. Eine weitere Einstellungsmoglichkeit ist im gleichen
Dialogfeld das Kontrollkistchen Ausgeblendet. Damit werden die eigentlichen Formeln
versteckt. Beide Schalter sind aber, wie schon gesagt, erst nach der Aktivierung des Tabel-
lenschutzes wirksam. Diesen Tabellenschutz werden wir gleich als Nichstes behandeln.

Die soeben beschriebenen manuell vorgenommenen Sperreinstellungen kénnen Sie auch
programmatisch festlegen, wie das Makro aus Listing 4.6 zeigt.

Sub TabelleVorbereiten()
Dim Zelle As Range

With tb1_FormelnUndWerte
For Each Zelle In .UsedRange

If Zelle.Font.Bold = True Then
Zelle.Locked = True
Zelle.FormulaHidden = True

Else
Zelle.Locked = False
Zelle.FormulaHidden = False

End If

131



Kapitel 4: Das Objekt Worksheet (Tabellen programmieren)

Next Zelle
End With

End Sub

Listing 4.6: Das Sperrmerkmal nur fiir Zellen mit Fettschrift einstellen.

Rein optisch sehen Sie nach dem Start des Makros aus Listing 4.6 nichts. Wenn Sie aber
das Dialogfeld Zellen formatieren fiir eine Zelle ohne Fettschrift iber die Tastenkombina-
tion +(1) aufrufen, kénnen Sie auf der Registerkarte Schutz erkennen, dass die Kon-
trollkdstchen Gesperrt und Ausgeblendet nicht aktiviert sind. Fithren Sie den Gegentest
auch einmal fiir eine Zelle mit Fettschrift durch. Dort miissen die Kontrollkistchen
Gesperrt und Ausgeblendet aktiviert sein.

Tabellenschutz fiir eine Tabelle einstellen und zuriicksetzen

Nachdem Sie eben entschieden haben, welche Zellen einer Tabelle gesperrt sein sollen,
konnen Sie nun damit beginnen, den Tabellenschutz fiir die Tabelle einzustellen. Im fol-
genden Makro aus Listing 4.7 wird ein Tabellenschutz fir die Tabelle tbl_FormelnUnd-
Werte eingestellt, und dabei wird das Kennwort Test festgelegt.

Sub TabellenSchutzAn()
tb1_FormelnUndWerte.Protect Password:="Test"

End Sub

Listing 4.7: Die Tabelle wird mit einem Passwort geschiitzt.

In diesem Einzeiler wenden wir die Methode Protect an und tbergeben im Parameter
Password das gerade definierte Passwort. Dass das hier festgelegte Passwort eher ungeeignet
ist, versteht sich von selbst. Dieses Passwort wird wahrscheinlich von jedem zuerst aus-
probiert. Da Sie das Passwort ja nicht hindisch erfassen miissen, konnen Sie sich Gedan-
ken machen, auch ein etwas anspruchsvolleres Passwort anzugeben.

Machen Sie sich aber dazu nicht zu viele Gedanken — egal wie lang und schwer das einge-
stellte Passwort auch ist — es ist eine Angelegenheit von wenigen Sekunden, das Passwort
aufzuhebeln. Dazu gibt es massenhaft Gratistools im Internet. Der Blattschutz von Excel
ist eher ein Schutz vor versehentlichem Uberschreiben von Daten, mehr nicht. Wenn Sie
einen zuverldssigen Schutz Threr Mappen und Tabellen wiinschen, kann ich Thnen die
Software LockXLS empfehlen, die ich selbst getestet und im Einsatz habe. Was jetzt noch
fehlt, ist ein Makro, das uns die gerade geschiitzte Tabelle wieder entsperrt.

Sub TabellenSchutzAus()
tb1_FormelnUndWerte.Unprotect Password:="Test"

End Sub

Listing 4.8: Den Tabellenschutz wieder aufheben.

Zum Aufheben des eingestellten Tabellenschutzes verwenden wir die Methode Unprotect
und tibergeben als Parameter das gleiche Kennwort wie in Listing 4.7.

Alle Tabellen einer Mappe schiitzen

Wenn Sie einmal alle Tabellen einer Arbeitsmappe hindisch geschiitzt haben, wissen Sie,
dass Sie diesen Vorgang wirklich Tabelle fiir Tabelle vornehmen miissen. Dabei miissen
Sie das Kennwort jeweils zweimal eingeben. Das ist ein irrer Aufwand, der tiber den Ein-
satz eines Makros auf weniger als eine Sekunde verkiirzt werden kann. Das Makro aus Lis-
ting 4.9 schiitzt alle Tabellen in der Mappe, wobei der Schutz hierbei etwas aufgeweicht
wird. Es soll dem Anwender moglich sein, trotz des eingestellten Blattschutzes noch Zel-
len zu formatieren.

Sub TabellenSchiitzen()
Dim Blatt As Worksheet

For Each Blatt In Worksheets
Blatt.Protect Password:="Test", AllowFormattingCells:=True
Next Blatt

End Sub

Listing 4.9: Alle Tabellen in einem Aufwasch schiitzen.

Zu Beginn deklarieren wir eine Objektvariable vom Typ Worksheet. Danach wenden wir
eine For Each...Next-Schleife an, um die Mappe Tabelle fir Tabelle abzuarbeiten. Innerhalb
der Schleife kommt die Methode Protect zum Einsatz. Uber den Parameter Password geben
wir das gewiinschte Kennwort an. Im Argument AllowFormattingCells legen wir fest, dass
der Anwender trotz eingestellten Blattschutzes Zellen formatieren darf.

Weitere Aufweichungen des Tabellenschutzes kénnen Sie der Syntax entnehmen, wenn Sie
die Methode Protect markieren und die Taste driicken. Die Syntax dazu lautet etwas
grob zusammengefasst:

Ausdruck.Protect (Password, DrawingObjects, Contents, Scenarios, UserInterfaceOnly,
AlTowFormattingCells, AllowFormattingColumns, AllowFormattingRows, AlTowInsertingColumns,
AllowInsertingRows, AllowInsertingHyperlinks, AlTowDeletingColumns, AllowDeletingRows,
AllowSorting, AllowFiltering, AllowUsingPivotTables)

Welche Bedeutung diese Schalter haben, konnen Sie eigentlich ganz einfach in Erfahrung
bringen. Wechseln Sie dazu zur normalen Excel-Oberfliche und klicken Sie im Menii-
band Uberpriifen auf Blatt schiitzen.

Blattschitzen | i (e
I Arbeitsblatt und Inhalt gesperrter Zellen schitzen I

Kennwort zum Aufheben des Blattschutzes:

Alle Benutzer dieses Arbeit dirfen:
Gesperrte Zellen auswihlen
Micht gesperrte Zellen auswahlen

Zellen formatieren

Spalten formatieren

Zeilen formatieren

Spalten einfiigen

Zeilen einfigen

Hyperlinks einfigen

Spalten laschen

Zeilen laschen &7

[ ok | [ abbrechen |

Abbildung 4.8: Die verfiigbaren Schalter beim Tabellenschutz anwenden.

Die deutschsprachigen Optionen koénnen mit den englischen Schaltern der Methode
Protect gut abgeglichen werden.

Das Makro zum Entsperren aller Tabellen sehen Sie in Listing 4.10.

Sub TabellenEntsperren()
Dim Blatt As Worksheet

For Each Blatt In Worksheets
Blatt.Unprotect Password:="Test"
Next Blatt

End Sub

Listing 4.10: Alle Tabellen wieder entsperren

Wir wenden die Methode UnProtect in einer Schleife an und iibergeben dieser Methode als
einzigen Parameter das Kennwort, um den Tabellenschutz aufzuheben.

9 Ein kleines Sicherheitsleck gibt es noch. Wechseln Sie einmal zurlick zur Standard-
AP oberflache von Excel und klicken Sie im Menlband Entwicklertools auf die Schaltfla-
che Makros.

Makro

Makroname:

TabellenSchutzus [ Ausfdhren ]
-

SVerweisErsatz
TabellenEntsperren Schritt
TabellenKopfEinflgen
TabellenKopfEinfOgenAktiveTabelle Bearbeiten
TabellenKopfEinflgenAktiveTabelleMitCells =
TabellenKopfEinflgenMitVorherigemLdschen
TabellenlistenUndInhaltsverzeichnisErstellen
TabellenListenUndInhaltsverzeichnisErstellenV2 D‘
TabellenSchutzAn

TabellenSchitzen Optionen...
TabelleVorbereiten S

Erstellen

Makros in: |Alle offenen Arbeitsmappen IZ|

Beschreibung

Abbrechen

Abbildung 4.9: Unser Makro TabellenSchutzAus wird hier angezeigt.

Da die Tabellenschutzmakros in diesem Fenster angeboten werden, kdnnen sie
auch von jedermann gestartet werden, was eher nicht so praktisch ist. Wie aber
kann verhindert werden, dass sensible Makros dort angezeigt werden? Ganz einfach
—indem Sie in den Kopf des jeweiligen Makros das Schlisselwort Private schreiben.

Private Sub TabellenSchiitzen()
Private Sub TabellenEntsperren()

Wird ein Makro als Private deklariert, wird es nicht im Makro-Dialogfeld angezeigt.

Tabellenblatter anlegen, drucken und exportieren

Anhand der nachfolgenden Beispiele lernen Sie, wie Sie Tabellen anlegen, drucken und
exportieren konnen.

Tabellen anlegen und benennen

Im nichsten Beispiel aus Listing 4.11 werden Sie eine neue Mappe anlegen und darin
genau zwolf Tabellen einfiigen. Diese sollen dann automatisch die Namen der zwolf
Monate erhalten.

Sub MappeAnlegenMonatsTabellenEinfiigen()
Dim intAnzahl As Integer

'Zundchst die eingestellte Anzahl von Tabellen aus dem System abfragen
intAnzahl = Application.SheetsInNewWorkbook

'dann diese Standardeinstellung tempordr &ndern
Application.SheetsInNewhWorkbook = 12

'Arbeitsmappe mit diesen 12 Tabellen einfiigen
Workbooks . Add

'und den urspriinglich eingestellten Wert wiederherstellen
Application.SheetsInNewhWorkbook = intAnzahl

For intAnzahl = 1 To 12
Worksheets (intAnzah1) .Name = MonthName (intAnzahl)
Next intAnzahl

End Sub

Listing 4.11: Eine neue Mappe anlegen, zwdélf Tabellen einfiigen und diese nach den Monaten benennen.

Zu Beginn des Makros deklarieren wir zunichst eine Variable vom Typ Integer mit dem
wabhlfreien Namen intAnzahl. Die Deklaration sorgt dafiir, dass im Arbeitsspeicher Platz
fiir unsere Variable reserviert wird. Aktuell hat diese Variable noch den Wert 0. Wir fiillen
die Variable jetzt, indem wir die Standardeinstellung fiir die Anzahl der Tabellen iiber die
Eigenschaft SheetsInNewhorkbook abfragen. Diese Eigenschaft ist unterhalb des hochsten
Objekts in Excel, also unter Application, angesiedelt. Wir iiberschreiben den standardma-
Rig eingestellten Wert nun mit den geforderten zwolf Tabellenblittern. Danach wenden
wir die Methode Add an, um eine neue Arbeitsmappe anzulegen. Am Ende des Makros
missen wir den vorher eingestellten Standard wiederherstellen. Dazu weisen wir der
Eigenschaft SheetsInNewhorkbook den Inhalt der Variablen intAnzahl zu.

Anschliefend verwenden wir die Variable intAnzahl zu einem anderen Zweck, nimlich zum
Zweck der Benennung der Tabellen. In einer For...Next-Schleife werden die einzelnen Tabel-
len nacheinander verarbeitet. In der Schleife geben wir den Tabellen die Namen der Monate
eines Jahres. Dazu weisen wir der Eigenschaft Name das Ergebnis der Funktion MonthName zu.
Diese Funktion erwartet einen Wert zwischen 1 und 12 und liefert Thnen dazu automatisch
den landesspezifischen Monatsnamen. Wenn Sie also in der Systemsteuerung ein anderes
Land, beispielsweise Polnisch, einstellen, gibt das Makro die Monatsnamen auf Polnisch
aus und benennt die Tabellen dementsprechend auch polnisch.

-

4« v .| Januar | Februar | Marz | April | Mai | Juni | Juli | August | e @ i [
Bereit 25 H O -——F——+ 100%

Abbildung 4.10: Die neue Mappe mit den zwélf Monaten wurde automatisch angelegt.

Eine Tabelle drucken

Fiir den normalen Ausdruck einer Tabelle steht Thnen die Methode PrintOut zur Verfii-
gung. Wenn es denn ein PDF sein soll, verwenden Sie die Methode ExportAsFixedFormat. Im
Beispiel aus Abbildung 4.11 wird die Tabelle tbl_Tabelle gedruckt. Dazu wird vorher
jedoch das Seitenformat noch festgelegt.

i Dotei Bearbeiten Ansicht Dinfogen Format Debuggen Ausfihwen [Extras  Add-fns Fenster 1 8w

HE-Wd s BRI 0 a K EFY # @ zmasu ] BowmaeEFraz2ianin]
 [Projekt - VBAProject 1% [1Angemeiny =] [raceneorucken = |
o =[] Sub TabelleDrucken() |
=] VBAPropect (Start.dsm) 1
) (] Meseroft Fxerd Dbkl 'Quesfors inatellen
B 55 Modkie : Pagedetup. Orientation = xllandscape
£ mel_FrateMakron
i md_Furkioren tbl_Tabellen.Printout
w2 md_Makrorskorder PrintOutl]Fromy, [ Toj, [Cape], [Proviow] Al ], [Covate, | . [rgn as}
w2 md_srachelemente End_Sub
w2 md_Tabelen
w2 md_Zeken
dl_Tabellen x|

md_Tebelen, J
ST o]

Abbildung 4.11: Der Ausdruck erfolgt iiber die Methode PrintOut.

Anhand der Quickinfo, die automatisch erscheint, wenn Sie die Methode PrintOut
geschrieben und die gedriickt haben, sehen Sie, dass diese Methode einige
Parameter besitzt, mit denen Sie bestimmen kénnen, was wo wie oft gedruckt werden soll.
Die einzelnen Parameter sind in der Onlinehilfe anschaulich beschrieben, sodass ich an
dieser Stelle nicht weiter darauf eingehen maochte.

Alle sichtbaren Tabellen einer Mappe ausdrucken

Im folgenden Beispiel aus Listing 412 sollen alle sichtbaren Tabellen einer Mappe
gedruckt werden.

Sub NurEingeblendeteTabellenDrucken()
Dim Blatt As Worksheet

For Each Blatt In ThisWorkbook.Worksheets

If Blatt.Visible = x1SheetVisible Then
Blatt.PrintOut

End If

Next Blatt

End Sub

Listing 4.12: Alle sichtbaren Tabellen einer Mappe drucken.

In einer For Each...Next-Schleife werden alle Tabellen, bei denen uns die Eigenschaft
Visible die Konstante x1SheetVisible meldet, mithilfe der Methode Printout ausgedruckt.

Tabelle als PDF ausgeben

Seit Excel-Version 2007 ist es standardmifig méglich, eine Tabelle direkt im PDF-Format
auszugeben. Im Makro aus Listing 4.13 wird die Tabelle tbl_Tabellen in ein PDF-Doku-
ment umgewandelt und nach der Erstellung am Bildschirm angezeigt.

Sub TabelleAlsPDFAusgeben()

tb1 Tabellen.ExportAsFixedFormat Type:=x1TypePDF, _
Quality:=x1QualityStandard, OpenAfterPublish:=True

End Sub

Listing 4.13: Eine Tabelle in ein PDF-Dokument umwandeln.

Bei der Methode ExportAsFixedFormat gibt es einige Parameter, die Sie verwenden konnen,
um beispielsweise die Druckqualitit einzustellen oder auch die Moglichkeit zu nutzen,
das PDF nach der Erstellung am Bildschirm anzuzeigen.

] Start.pdf - Adobe Reader — a PR

Datei Bearbeiten Anzeige Fenster Hilfe E3

2 & @ | ) (8) fl ‘ | lz‘ ¥ Werkzeuge Kommentar

Iester
Tabellel 1
1bl ErsteMakros
Tabelle2

bl Kopf

miE

thl Scheife I
1bl_Datumsreihe I
tbl_Duplikate

bl ZeilenEntfernen
bl Artikel

bl Spalten

1bl SpaltenEntfernen
1bl Preise

tbl Zellen

bl Konvertieren

bl Suchen

bl Direktsuche

Abbildung 4.12: Das gerade erstellte PDF-Dokument wird automatisch gedffnet.

Eine Tabelle exportieren

Leider gibt es keinen gleichnamigen Befehl fiir den Export von Tabellen. Um eine Tabelle
beispielsweise in eine neue Arbeitsmappe zu exportieren, wird die Methode Copy, wie in
Listing 4.14 gezeigt, verwendet.

Sub TabelleExportieren()
tb1_Tabellen.Copy

End Sub

Listing 4.14: Ein einfacher Copy-Befehl reicht aus, um eine Tabelle zu exportieren.

Individuelle Kopf- und FuBzeilen erstellen

Im folgenden Makro soll eine individuelle Kopf- und FuRzeile fiir alle Tabellen einer
Arbeitsmappe erstellt werden. Da wir ja mehr konnen wollen als der normale Anwender
in Excel, haben wir gerade fur diese Aufgabenstellung zusitzliche Anforderungen einge-
baut, die mit normalen Bordmitteln nicht méglich sind. Entnehmen Sie die Aufgabenstel-
lung fiir dieses Beispiel direkt dem Makro aus Listing 4.15.

'Aufgabe: Individuelle Kopf- und FuBzeilen einrichten

'Kopfzeile, Tinks --> Anwendername 1t. Windows-Anmeldung, Computernamen
'Kopfzeile, Mitte --> Firmenname

'Kopfzeile, rechts --> Datum (Format: 12.03.16)

'"FuBzeile, links --> Erstelldatum & letzte Anderung (untereinander)
'FuBzeile, Mitte --> Leer

'FuBzeile, rechts --> Pfad, inkl. Dateinamen

Sub KopfUndFuBzeileEinstellen()
Dim Blatt As Worksheet
Dim strFirma As String

'Abfragen, ob die Dokumenteigenschaft Firma lberhaupt gesetzt ist

If ActiveWorkbook.BuiltinDocumentProperties("company") = "" Then

'Wenn nicht, dann Firma lber InputBox abfragen

strFirma = InputBox("Bitte den Firmennamen eingeben", "Firma angeben")
ActiveWorkbook.BuiltinDocumentProperties("company") = strFirma
Else

strFirma = ActiveWorkbook.BuiltinDocumentProperties("company")
End If

For Each Blatt In Worksheets
With Blatt.PageSetup

.LeftHeader = Environ("Username") & " / " & Environ("Computername")

.CenterHeader = strFirma

.RightHeader = Format(Date, "DD.MM.YY")

.LeftFooter = "Erstelldatum: " & _

Format (ActiveWorkbook.BuiltinDocumentProperties("creation date"), "DD.MM.YY") & vbLf & _
"Letzte Anderung: " & _
Format (ActiveWorkbook.BuiltinDocumentProperties("last save time"), "DD.MM.YY") & _
vbLf & ActiveWorkbook.FullName

.CenterFooter = "" 'Loscht einen evtl. gesetzten Eintrag

.RightFooter = ""

End With
Next Blatt

End Sub

Listing 4.15: Eine individuelle Kopf- und Fu/fSzeile zusammenstellen.

Der Firmenname kann tiber die Dokumenteigenschaft BuiltinDocumentProperties ("company")
abgefragt werden. Wenn diese standardmiiRig nicht gesetzt ist, wird der Anwender dazu
aufgefordert, seinen Firmennamen einzugeben. Der so eingegebene Firmenname wird im
Anschluss direkt in die Dokumenteigenschaft geschrieben und dann iiber die Eigenschaft
CenterHeader in die Mitte der Kopfzeile platziert.

Uber die Eigenschaft LeftHeader konnen wir auf den linken Bereich der Kopfzeile zugrei-
fen. Dort wenden wir die Funktion Environ an, iiber die wir mittels der Konstanten Username
und Computername den Namen des angemeldeten Users sowie den Namen des Computers
abfragen. Uber die Eigenschaft RightHeader haben wir Zugriff auf den rechten Teil der
Kopfzeile. Dort schreiben wir ein formatiertes Datum hinein. Das Datum erhalten wir
direkt aus der Funktion Date, das gewiinschte Format stellen wir mithilfe der Funktion
Format ein.

Fiir die FuRzeile setzen wir die Eigenschaft LeftFooter ein, um Zugriff auf den linken
Bereich der Fulzeile zu erhalten. Dort sollen einige Zeilen untereinandergeschrieben wer-
den. Den Zeilenumbruch bekommen wir mithilfe der Konstanten vbLf hin. Die Informati-
onen dariiber, wann die Mappe angelegt und wann die letzte Anderung daran registriert
wurde, erhalten wir iiber die Konstanten creation date und Last save time, die wir direkt
aus den Dokumenteigenschaften beziehen. Den Eigenschaften CenterFooter und
RightFooter weisen wir einen sogenannten Leerstring zu. Damit wird gleichzeitig ein even-
tuell vorher dort stehender Text geloscht.

Bernd.Held / ENTWICKLUNG Held-Office 19.11.15
Fall 1 Fall 2 Fall 3 Fall4 Fall 5 Fall&

250 Held 123-456-56 123-45/67 129,13- 26.022016
45,99 Held 123-456-57 123-45/68 12,467- 13.10.13
467,89 Held 123-456-58  123-45/69 123,67 1.0kt2016
53,99 Schmitt 123-456-59  123-45/70 532,567- 26.022016
-456 Fuchs 123-456-60 123-45/71 45,5 27.022016
9,99 Krieger 123-456-61 123-45/72 28.02.2016
Test Bernd Held ~ 123-456-62  123-45/73 01.02.2016

Abbildung 4.13: Die Kopfzeile ...

Erstelldatum:02.03.13

Letzte Anderung: 19.11.15

‘\Hpserver\webdaten\Publikationen\Bicher\DPunkt\Richtig einsteigen-Excel-VBA 2016%Dateien zum
Buch\Startxlsm

Abbildung 4.14: ... sowie die dazugehirige FufSzeile.

Tabellen verstecken oder loschen

Auf den folgenden Seiten beschreibe ich, wie Sie Tabellen ein- und ausblenden sowie ein-
zelne Tabellen 16schen konnen.

Tabellen ein- und ausblenden

Erinnern Sie sich an den Anfang des Buchs? Dort haben wir gemeinsam tiber das Eigen-
schaftenfenster bestimmte Tabellen ein- und ausgeblendet, indem wir der Eigenschaft
Visible eine Konstante zugewiesen haben. Im folgenden Makro aus Listing 4.16 wird die
Tabelle tbl_Tabellen ein- und wieder ausgeblendet.

Sub TabelleEinUndAusblenden()

If tbl Tabellen.Visible = x1SheetVisible Then
tb1 Tabellen.Visible = x1SheetHidden

Else
tb1 Tabellen.Visible = x1SheetVisible

End If

End Sub

Listing 4.16: Ein Makro, das beides kann.

Im Makro aus Listing 4.16 wird gepriift, wie der augenblickliche Status der Tabelle ist. Ist
die Tabelle sichtbar, wird sie ausgeblendet. Ist sie ausgeblendet, wird sie eingeblendet.

Alle Tabellen bis auf eine ausblenden

Im folgenden Makro aus Listing 4.17 werden alle Tabellen mit Ausnahme der Tabelle
tbl_Tabellen ausgeblendet.

Sub AlleTabellenBisAufEineAusblenden()
Dim Blatt As Worksheet

For Each Blatt In Worksheets
Select Case Blatt.CodeName
Case "tb1 _Tabellen"

Case Else
Blatt.Visible = x1SheetVeryHidden

End Select
Next Blatt

End Sub

Listing 4.17: Alle Tabellen bis auf die Tabelle tbl_Tabellen werden ausgeblendet.

Auch fiir diese Aufgabe konnen wir die For Each...Next-Schleife verwenden, um alle Tabel-
len nacheinander abzuarbeiten. Innerhalb der Schleife arbeiten wir mit der Select Case-
Struktur und fragen den Codenamen der Tabelle iiber die entsprechende Eigenschaft ab.

Das Makro bleibt nur dann dauerhaft lauffihig, wenn Sie auf den Codenamen der Tabelle
zugreifen, auf den der normale Anwender keinen Zugriff hat. Das bedeutet, dass der Stan-
dardanwender die Tabellen benennen kann, wie er mochte, und Thr Makro wird trotzdem
weiterhin wie gewiinscht ausgefiithrt. Der Case Else-Zweig wird fur alle Tabellen durchlau-
fen, deren Codename nicht tbl_Tabellen lautet. Diese Tabellen werden iiber die Konstante
x1SheetVeryHidden, die Sie der Eigenschaft Visible zuweisen, auf sichere Art und Weise aus-
geblendet.

i Mdchten Sie noch weitere Tabellen im eingeblendeten Status erhalten, erweitern
A Sie den ersten Case-Zweig um weitere Tabellennamen, jeweils getrennt voneinan-
der durch ein Komma und eingeschlossen in doppelte Anflihrungszeichen zu
Beginn und am Ende des jeweiligen Tabellennamens.

Case "tbl _Tabellen", "tbl _Eingabe", "tbl Schleife"

In diesem Fall wiirden diese drei Tabellen nicht ausgeblendet werden.

Um die gerade ausgeblendeten Tabellen wieder sichtbar zu machen, wenden Sie das
Makro aus Listing 4.18 an.

Sub AlleTabellenAnzeigen()
Dim Blatt As Worksheet

For Each Blatt In Worksheets
Blatt.Visible = x1SheetVisible
Next Blatt

End Sub

Listing 4.18: Alle Tabellen nacheinander mithilfe einer Schleife wieder sichtbar machen.

In der Schleife weisen wir der Eigenschaft Visible die Konstante x1SheetVisible zu, um die
Tabellen wieder einzublenden.

Tabellen loschen

Im folgenden Beispiel sollen einige Tabellen aus der Arbeitsmappe entfernt werden. Da
bereits bestehende Beispieltabellen nicht geléscht werden sollen, fiigen wir zur Laufzeit
des Makros eine neue Tabelle hinzu, die dann gleich wieder geloscht wird.

Sub TabelleEinfiigenUndWiederLdschen()

'Zuerst eine neue Tabelle am Ende der Mappe einfiigen
Worksheets.Add after:=Worksheets(Worksheets.Count)

'Riickfrage vor dem Loschen der Tabelle tempordr ausschalten
Application.DisplayAlerts = False

'Tabelle 16schen
ActiveSheet.Delete

'Riickfrage von Excel wieder einschalten
Application.DisplayAlerts = True

End Sub

Listing 4.19: Eine Tabelle am Ende der Mappe einfiigen und wieder léschen.

Damit Sie tiberhaupt etwas sehen, arbeiten Sie das Makro aus Listing 4.19 bitte Schritt fur
Schritt im Einzelschrittmodus durch (Taste (Fg]). Mithilfe der Methode Add fiigen wir der
Worksheets-Auflistung eine weitere Tabelle am Ende der Mappe hinzu. Um die letzte
Tabelle der Mappe zu ermitteln, verwenden wir die Funktion Count. Beim Versuch, eine
Tabelle zu loschen, meckert Excel normalerweise und fragt Sie, ob Sie die Tabelle wirklich
16schen mochten.

Microsoft Excel

i Das Ldschen von Blattern 13sst sich nicht rickgéngig machen, und maglicherweise entfernen Sie dabei Daten. Wenn Sie die Daten nicht bendtigen, klicken Sie auf “Loschen”.

[ Laschen ] [ Abbrechen ]

Abbildung 4.15: Die Riickfrage erscheint automatisch beim Versuch, eine Tabelle zu entfernen.

Diese Riickfrage konnen wir bei der Programmierung von Makros tiberhaupt nicht
gebrauchen, weil wir ja genau wissen, was wir tun. Daher schalten wir diese Riickfragen
iiber das Anwendungsobjekt (Application) direkt ab, indem wir der Eigenschaft
DisplayAlerts den Wert False zuweisen.

Zusammenfassung

Sie haben in diesem Kapitel gelernt, wie Sie mithilfe des Objekts Worksheet Tabellen pro-
grammieren kénnen. Das Objekt Worksheet wird dabei sehr oft mit der For Each...Next-
Schleife verwendet.

In Kapitel 5, »Das Objekt Workbook (Arbeitsmappen programmieren)«, gehen wir in der
Objekthierarchie von Excel wieder eine Ebene nach oben. Uber dem Objekt Worksheet
befindet sich das Objekt Workbook.

Die Lernkontrolle

Zum Abschluss dieses Kapitels stelle ich Thnen ein paar Verstindnisfragen. Die Antworten
auf die hier gestellten Fragen finden Sie im Anhang dieses Buchs.

*  »Welche Risiken birgt die Anweisung Sheets (1) .Range("A1").Value = 1007«
*  »Welche drei Konstanten weist die Eigenschaft Visible auf?«

* »Uber welche Eigenschaft kénnen Sie die Anzahl der Tabellen fiir eine neue Mappe
festlegen?«

* »Wie heiflt die Methode, um eine Tabelle in eine neue Mappe zu exportieren?«
* »Wie konnen Sie verhindern, dass Makros im Makro-Fenster angezeigt werden?«
* »Wie konnen Sie das Anlagedatum einer Mappe abfragen?«

*  »Wie heiflt der Schalter, den Sie verwenden, um Riickfragen in Excel temporir lahmzu-
legen?«



